

Scilab à l'agreg

Un exemple de programmation

21 octobre 2021

Le but de ces quelques pages est de proposer, au travers d'un exemple simple mais représentatif, une manière de programmer efficacement lors de l'épreuve de modélisation. L'idée générale est de commencer par des codes minimaux, qui sont enrichis par la suite. À chaque étape, on teste le programme, ce qui réduit de beaucoup les risques d'erreur dans le code.

1 Contexte

On va montrer comment mettre en œuvre efficacement la méthode d'Euler pour la résolution d'équations différentielles. Pour une équation du type

$$y'(t) = f(t, y(t)),$$

elle consiste – pour un pas h donné – à construire une suite d'approximations (y_n) de la solution y aux temps $t_n = nh$, par la formule

$$y_{n+1} = y_n + hf(t_n, y_n).$$

L'initialisation est fournie par la donnée de la condition initiale $y(0) = y_0$. Par la suite, on utilisera l'exemple modèle $y' = t - ty$ avec la condition initiale $y(0) = 2$, dont la solution exacte est donnée par $y(t) = 1 + \exp(-t^2/2)$.

La fonction scilab correspondante est la suivante :

```
1  function yp=f(t,y)
2  yp=t-t*y;
3  endfunction
```

N.B. (Gestion des fichiers) L'usage dans scilab est de placer les fonctions dans des fichiers suffixés `.sci` et les scripts dans des fichiers suffixés `.sce` (plusieurs fonctions peuvent être définies dans un même fichier `.sci`). Afin de pouvoir utiliser ces fonctions, il faut les placer préalablement en mémoire dans scilab par la commande `exec` suivie du nom du fichier correspondant. Cette même commande sert également à exécuter les fichiers de scripts.

2 Programmation hiérarchique

Une manière efficace de programmer en temps limité consiste à débiter par une version très épurée de l'algorithme enrichie ensuite, ceci en testant pas-à-pas chaque modification. Le débogage est ainsi grandement facilité, et si le temps imparti pour la programmation est terminé, on est sûr de pouvoir présenter au moins une simulation convaincante!

À chaque étape importante, on peut par exemple copier et commenter la précédente version du code afin de pouvoir revenir en arrière. Dans cet énoncé, la même fonction `Euler` aura trois versions.

2.1 Le cœur du programme

L'itération de la méthode d'Euler s'écrit simplement $y=y+h*f(t,y)$ si bien qu'une première version élémentaire de la méthode d'Euler est la suivante :

```
1 function y=Euler(y0,N,T)
2 // Version 1
3 y=y0;t=0;
4 h=T/N;
5 for i=1:N
6     y=y+h*f(t,y);
7     t=t+h;
8 end
9 endfunction
```

N.B. on a préféré ici passer le nombre de points N comme argument plutôt que le pas h afin d'éviter l'utilisation d'une partie entière.

2.2 Renvoi des arguments

Bien sûr, l'appel de la fonction précédente $y=Euler(y_0,N,T)$ ne fournit que l'approximation finale de $y(T)$, sans les valeurs intermédiaires... c'est un peu dommage si l'on souhaite tracer la solution. On y remédie en renvoyant la liste complète plutôt que la dernière valeur :

```
1 function liste_y=Euler(y0,N,T)
2 // Version 2
3 y=y0;liste_y=[y0];
4 t=0;
5 h=T/N;
6 for i=1:N
7     y=y+h*f(t,y);
8     t=t+h;
9     liste_y=[liste_y,y];
10 end
11 endfunction
```

Il peut être aussi commode que la fonction renvoie les temps successifs où sont effectuées les approximations :

```
1 function [liste_y,liste_t]=Euler(y0,N,T)
2 // Version 3
3 y=y0;liste_y=[y0];
4 t=0;liste_t=[0];
5 h=T/N;
6 for i=1:N
7     y=y+h*f(t,y);
8     t=t+h;
9     liste_y=[liste_y,y];
10    liste_t=[liste_t,t];
11 end
12 endfunction
```

Il est évident que le vecteur `liste_t` retourné vaut aussi bien `linspace(0,T,N+1)`, mais dans le cas d'une méthode à pas variable, il sera plus facile d'adapter la construction proposée.

2.3 Programme principal – appel de la fonction

On considère la fonction `Euler`, Version 3. Si on l'appelle – en ligne de commande Scilab – avec un seul argument de sortie, elle renverra le premier :

```
--> sol = Euler(2,5,2)

sol =

    2.    2.    1.84    1.5712    1.297024    1.1069286
```

Si l'on souhaite avoir accès aux deux arguments de sortie, il faut effectuer l'appel comme suit :

```
--> [sol, tps] = Euler(2,5,2)

tps =
```

```

0.    0.4    0.8    1.2    1.6    2.

sol  =

2.    2.    1.84    1.5712    1.297024    1.1069286

```

N.B. La syntaxe scilab est quelque-peu ambiguë en ce qui concerne les arguments de sortie. En effet, les crochets `[sol, tps]` n'ont ici rien d'un assemblage matriciel, les objets `sol` et `tps` n'ont aucune raison d'avoir des tailles compatibles, ni même des types identiques.

La possibilité de renvoyer à la fois les temps et les valeurs des approximations permet une utilisation très simple de la fonction :

```

1 // Parametres
2 T=2;
3 N=50;
4 y0=2;
5 // Calcul
6 [sol, tps]=Euler(y0, N, T);
7 // Graphique
8 plot(tps, sol)

```

2.4 Bonnes pratiques

Si l'on souhaite comparer l'approximation obtenue avec la solution exacte, on peut définir une fonction :

```

1 function y=yex(t)
2 y=1+exp(-t.^2/2);
3 endfunction

```

Noter l'utilisation de l'opérateur `.` afin d'élever terme-à-terme un vecteur au carré. Elle permet un appel unique pour l'évaluation de la solution exacte sur la subdivision :

```

1 plot(tps, sol)
2 plot(tps, yex(tps), 'r')

```

Par ailleurs, la méthode d'Euler programmée est indépendante de la dimension, pour peu qu'on impose aux vecteurs d'être écrits en colonne. Par exemple, pour résoudre le système différentiel suivant :

$$\begin{cases} x'(t) = x(t)(3 - y(t)), \\ y'(t) = y(t)(-2 + 2x(t)), \end{cases}$$

il suffit de redéfinir la fonction comme suit

```

1 function yp=f(t, y)
2 yp=[y(1)*(3-y(2)); y(2)*(-2+2*y(1))];
3 endfunction

```

et le programme principal suivant trace les trajectoires en fonction du temps, ainsi que dans le plan de phase :

```

1 // Parametres
2 T=10;
3 N=5000;
4 y0=[1; 1];
5 // Calcul
6 [sol, tps]=Euler(y0, N, T);
7 // Graphique
8 subplot(2,1,1)
9 plot(tps, sol(1,:), tps, sol(2,:))
10 title('Solutions x(t) et y(t)')
11 subplot(2,1,2)
12 plot(sol(1,:), sol(2,:))
13 title('Plan de phase')

```

Insistons enfin sur le fait que la fonction Euler ne trace aucune courbe. Il est préférable de dissocier le post-traitement graphique du calcul.

3 Raffinements

3.1 Habillage graphique

Lors de l'oral, il est important, pour soi et pour le jury, que les graphes portent quelques indications qui permettent d'identifier les données : titre, légendes, axes, etc.

```
1 // Parametres
2 T=2;
3 N=50;
4 y0=2;
5 // Calcul
6 [sol, tps]=Euler(y0, N, T);
7 // Graphique
8 close()
9 plot(tps, sol, 'thicknes', 2)
10 plot(tps, yex(tps), 'r--', 'thicknes', 2)
11 // Titres et legendes
12 title('Resolution par la methode d''Euler', ...
13       'fontsize', 3)
14 xlabel('temps t', 'fontsize', 3)
15 ylabel('y(t)', 'fontsize', 3)
16 legend('Solution numerique', 'Solution exacte')
```

N.B. Noter l'utilisation des trois points pour écrire sur deux lignes une commande très longue.

3.2 Erreur d'approximation – Ordre de convergence

Si l'on souhaite mettre en évidence la convergence d'ordre 1 de la méthode d'Euler, on doit effectuer des simulations pour différentes valeurs du pas h . Il suffit d'ajouter une boucle extérieure à notre programme principal.

```
1 // Parametres
2 T=2;
3 y0=2;
4 liste_N=10:10:1000;
5 // Boucle sur N
6 liste_Err=[];
7 for N=liste_N
8     // Calcul
9     [sol, tps]=Euler(y0, N, T);
10    // Evaluation de l'erreur
11    erreur=norm(sol-yex(tps), 'inf');
12    // Mise a jour du tableau d'erreurs
13    liste_Err=[liste_Err, erreur];
14 end
15 // Trace
16 close()
17 plot(liste_N, liste_Err, 'o-')
18 title('Erreur en fonction de N')
19 xlabel('Nombre de points N')
20 ylabel('Erreur uniforme')
21
22 xclick();
23
24 clf
25 plot2d(liste_N, liste_Err, logflag='ll')
26 title('Erreur en fonction de N (log-log)')
27 xlabel('Nombre de points N')
28 ylabel('Erreur uniforme')
29
30 xclick();
31
32 [a, b, sig]=reglin(log(liste_N), log(liste_Err))
33 chaine=['Pente=', string(a)];
34 xstring(100, 1e-3, chaine)
```

Ce programme peut paraître long et assez complexe, mais il est le résultat de suites de programmes qui diffèrent les uns des autres de quelques lignes seulement. Comme tous les codes intermédiaires ont été testés et validés, il est peu probable que le code final contienne des erreurs.

3.3 Passage de fonction comme paramètre

On a vu plus haut qu'à chaque nouvelle équation différentielle, la fonction `f.m` devait être modifiée. Il peut être plus commode de définir une fonction par équation, et permettre un choix dans la fonction `Euler`. La fonction est alors passée en paramètre.

```

1  function [liste_y, liste_t]=Euler(f,y0,N,T)
2  // Version 4
3  y=y0; liste_y=[y0];
4  t=0; liste_t=[0];
5  h=T/N;
6  for i=1:N
7      y=y+h*f(t,y);
8      t=t+h;
9      liste_y=[liste_y,y];
10     liste_t=[liste_t,t];
11 end
12 endfunction

```

L'appel s'effectue alors par

```

--> sol = Euler(f,2,5,2)

sol =

    2.    2.    1.84    1.5712    1.297024    1.1069286

```

3.4 Améliorer la portabilité de vos programmes

Il peut arriver de lancer son programme dans une autre session de scilab, disons sur l'ordinateur de la salle d'épreuve le jour J, et de constater avec désarroi un comportement inattendu (erreur ou résultats différents). La raison en est probablement qu'au fil de l'élaboration de vos programmes, certaines variables étaient restées en mémoire dans la session précédente sans être correctement chargées dans vos programmes. Pour se prémunir de cette difficulté, il est recommandé d'effacer les variables en début de script et de charger systématiquement les fichiers de fonctions utilisés.

```

1  clear;
2  exec f.sci;
3  exec Euler.sci;
4  ...

```

4 Exercices

4.1 Méthode de Newton

La méthode de Newton permet d'approcher la solution d'une équation

$$F(x) = 0,$$

où $F : \mathbb{R}^d \rightarrow \mathbb{R}^d$ est différentiable. Partant d'une approximation initiale $x_0 \in \mathbb{R}^d$ de la solution, la suite (x_k) est construite par la récurrence¹

$$x_{k+1} = x_k - [F'(x_k)]^{-1} F(x_k).$$

Programmer une fonction scilab selon l'entête suivant,

```
function x=newton(F,DF,x0,Tol,MaxIter)
```

qui renvoie une approximation de la solution x . Les arguments d'entrée sont les suivants :

- F : nom de la fonction scilab définissant la fonction F ,
- DF : idem pour la matrice jacobienne F' ,
- x_0 : vecteur initial $x^0 \in \mathbb{R}^d$,
- Tol : scalaire donnant la tolérance (critère d'arrêt sur l'incrément : $\|x^{k+1} - x^k\| < Tol$),
- $MaxIter$: nombre maximal d'itérations (pour les situations de non-convergence).

N.B. Pour gérer le critère d'arrêt², on utilisera une boucle `while`, selon le modèle suivant :

```
err = Tol+1;
iter = 0;
while (err>Tol)&(iter<MaxIter)
iter = iter +1;
...
end
```

Tester cette fonction sur l'exemple monodimensionnel simple $F(x) = e^x - e$, ainsi que sur le cas bi-dimensionnel

$$F(x_1, x_2) = \begin{bmatrix} x_1^2 + x_2^2 - 2 \\ x_1^2 - x_2^2 - 1 \end{bmatrix}.$$

4.2 Méthode d'Euler implicite

À l'aide de la fonction précédente, programmer une fonction scilab qui met en œuvre la méthode d'Euler implicite (ou *rétrograde*), dont l'itération s'écrit

$$y_{n+1} = y_n + hf(t_{n+1}, y_{n+1}).$$

N.B. La fonction F qui définit le système à résoudre à chaque itération dépend de y_n et de h , on écrira une variante de la fonction `newton` précédente pour prendre en compte ces deux arguments supplémentaires.

Comparer la méthode d'Euler explicite et la méthode d'Euler implicite sur le problème raide

$$y'(t) = -500(y(t) - \cos(t)),$$

ainsi que sur le système de Volterra-Lotka :

$$\begin{cases} x'(t) = ax(t) - bx(t)y(t), \\ y'(t) = -cy(t) + dx(t)y(t). \end{cases}$$

1. Remarque : pour les systèmes ($d > 1$), on évite de calculer la matrice inverse $[F'(x_k)]^{-1}$, mais on résout directement le système linéaire correspondant. Ainsi, on préfère écrire `DF(x)\F(x)` plutôt que `inv(DF(x))*F(x)`. Notez que matlab effectue automatiquement le remplacement, mais pas scilab.

2. Un autre exemple de critère d'arrêt possible est ($\|x^{k+1} - x^k\| > \|x^k - x^{k-1}\|$ ou $\|x^{k+1} - x^k\| = 0$). Ce critère permet d'atteindre la précision de la machine (donc sans utiliser de tolérance).

On pourra aussi programmer la méthode de Crank-Nicolson :

$$y_{n+1} = y_n + \frac{h}{2} [f(t_n, y_n) + f(t_{n+1}, y_{n+1})].$$

et proposer une estimation numérique de la période des oscillations.

4.3 Une méthode de tir

On considère le problème aux limites suivant :

$$\begin{cases} -u''(x) + a(x)u(x) = f(x), \\ u(0) = u(1) = 0, \end{cases}$$

où les fonctions $a \geq 0$ et f sont données. Pour résoudre ce problème, on introduit le problème de Cauchy dépendant du paramètre α :

$$\begin{cases} -u_\alpha''(x) + a(x)u_\alpha(x) = f(x), \\ u_\alpha(0) = 0 \text{ et } u_\alpha'(0) = \alpha, \end{cases}$$

Ce dernier problème peut être résolu à l'aide d'une méthode d'intégration des équations différentielles ordinaires vues plus haut. Il suffit de trouver α pour que $u_\alpha(1) = 0$, ce qui n'est pas difficile puisque l'application $\varphi : \alpha \mapsto u_\alpha(1)$ est affine !

Pour aller plus loin, on pourra appliquer la même méthode au problème

$$\begin{cases} -u''(x) + a(x)u^3(x) = f(x), \\ u(0) = u(1) = 0, \end{cases}$$

pour lequel l'application $\varphi : \alpha \mapsto u_\alpha(1)$ est non-linéaire. On pourra alors utiliser la méthode de Newton ou une variante pour résoudre le problème de tir.